



Quadriques

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

La première version de ce document est parue en janvier 2006. L'objectif principale de ce document est de présenter les quadriques dans une démarche de découvertes pas à pas. Les quadriques sont des surfaces remarquables de l'espace. La présentation de chaque quadrique est faite pas à pas à l'aide de traces laissées sur ces surfaces par des plans parallèles aux trois plans de coordonnées. Afin de saisir rapidement les caractéristiques de ces traces, l'élève doit avoir préalablement une connaissance des équations cartésiennes des coniques (ellipses, paraboles et hyperboles). Pour chaque quadrique, une illustration de ce à quoi la surface ressemble sera donnée en superposant dans un même graphique les différentes traces laissées par des plans parallèles aux plans de coordonnées. Ces traces auront été obtenues avec une utilisation judicieuse de la macro-commande `spacecurve`. C'est seulement après une perception du quadrique à l'aide des traces que l'on tracera directement la surface avec la macro-commande `implicitplot3d` pour les quadriques définis implicitement ou avec la macro-commande `plot3d` lorsque le quadrique est défini explicitement. La présentation de chaque quadrique se termine en donnant des équations paramétriques le décrivant. Comme vous le constaterez, avec de telles équations, le rendu de la surface pour la plupart des quadriques sera d'un rendu supérieur à celui donné avec la macro-commande `implicitplot3d`. Une lecture attentive de ce document permettra au lecteur de développer sa perception spatiale et d'augmenter ses compétences dans l'utilisation du logiciel Maple.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;
> with(plots,display,implicitplot,implicitplot3d,spacecurve,
  setoptions3d):
with(plottools,point):
macro(grispâle = COLOR(RGB, .9523, .9523, .9523)):
setoptions3d(size=[400,400],labels=[x,y,z],axesfont=[TIMES,ROMAN,14],
  tickmarks=[0,0,0],axis=[color="Niagara 4"],
  labelfont=[TIMES,ROMAN,15],titlefont=[TIMES,ITALIC,14],
  axes=normal,scaling=constrained);
```

L'initialisation suivante permettra d'avoir plus de lisibilité des nombres décimaux en supprimant les zéros non significatifs à la fin d'un nombre.

```
> interface(typesetting=extended); # Pour s'assurer le niveau de
  composition étendue
Typesetting:-Settings(striptrailing=true);
                                extended
                                false
```

(1.1)

Ellipsoïde d'équation $\frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} = 1$

Soit l'ellipsoïde d'équation $\frac{x^2}{16} + \frac{y^2}{4} + \frac{z^2}{4} = 1$.

```
> unassign('a,b,c');
```

```
Éq_générale:=x^2/a^2+y^2/b^2+z^2/c^2 = 1;
```

$$\text{Éq_générale} := \frac{x^2}{a^2} + \frac{y^2}{b^2} + \frac{z^2}{c^2} = 1 \quad (2.1)$$

```
> a:=4:
```

```
b:=2:
```

```
c:=2:
```

```
Ellipsoïde:=Éq_générale;
```

$$\text{Ellipsoïde} := \frac{x^2}{16} + \frac{y^2}{4} + \frac{z^2}{4} = 1 \quad (2.2)$$

Obtenons quelques les traces laissées sur cet ellipsoïde par des plans parallèles aux trois plans de coordonnées.

Commençons avec différents plans parallèles au plan de coordonnées xOz . De tels plans de l'espace ont pour équations une équation de la forme $y = k$. Posons alors $y = k$ dans l'équation de l'ellipsoïde.

```
> Éq_particulière:=subs(y=k,Éq_générale);
```

$$\text{Éq_particulière} := \frac{x^2}{16} + \frac{k^2}{4} + \frac{z^2}{4} = 1 \quad (2.3)$$

```
> Éq_transition:=Éq_particulière+(-k^2/4=-k^2/4);
```

$$\text{Éq_transition} := \frac{x^2}{16} + \frac{z^2}{4} = 1 - \frac{k^2}{4} \quad (2.4)$$

Nous constatons que, dans le plan $y = k$, pour chaque valeur k telle que $-2 \leq k \leq 2$, l'équation précédente représente une équation d'une ellipse ou d'un point. Exprimons l'équation de cette ellipse sous la forme réduite.

```
> Équation_ellipse:=(1/(1-1/4*k^2))*Éq_transition;
expand(Équation_ellipse);
```

$$\text{Équation_ellipse} := \frac{\frac{x^2}{16} + \frac{z^2}{4}}{1 - \frac{k^2}{4}} = 1$$

$$\frac{x^2}{16 \left(1 - \frac{k^2}{4}\right)} + \frac{z^2}{4 \left(1 - \frac{k^2}{4}\right)} = 1 \quad (2.5)$$

Obtenons les tracés de quelques ellipses. Nous allons utiliser des équations paramétriques d'une ellipse pour les tracer efficacement.

Des équations paramétriques d'une ellipse centrée à l'origine d'équation $\frac{x^2}{a^2} + \frac{y^2}{b^2} = 1$ sont

$$x = a \cos(t)$$

$$y = b \sin(t)$$

Posons alors, pour certains k dans l'ellipse d'équation $\frac{x^2}{16 \left(1 - \frac{k^2}{4}\right)} + \frac{z^2}{4 \left(1 - \frac{k^2}{4}\right)} = 1$,

$$a1 = 4 \sqrt{1 - \frac{k^2}{4}} \text{ et } b1 = 2 \sqrt{1 - \frac{k^2}{4}}$$

```
> a1:=4*sqrt(1-k^2/4);
  b1:=2*sqrt(1-k^2/4);
```

$$a1 := 2\sqrt{-k^2 + 4}$$

$$b1 := \sqrt{-k^2 + 4}$$

(2.6)

Construisons quelques ellipses avec des valeurs $-2 \leq k \leq 2$.


```
> #Pas:=[seq(-2+k*0.1,k=0..5),seq(-1+k*0.5,k=0..4),seq(1+k*0.1,k=5..10)
  ];
```

```
Pas:=[-2,-1.9,-1.7,-1.5,-1.3,-1.1,0,1.1,1.3,1.5,1.7,2];
```

```
Pas := [-2, -1.9, -1.7, -1.5, -1.3, -1.1, 0, 1.1, 1.3, 1.5, 1.7, 2]
```

(2.7)

```
> Ellipse_xOz:=seq(spacecurve([a1*cos(t),k,b1*sin(t)],t=0..2*Pi,color=
  "Bright 3"),k=Pas);
  display(Ellipse_xOz,view=[-5..5,-3..3,-3..3]);
```



Pour une meilleure perception spatiale de ce graphique 3d, cliquez sur l'image avec un clic gauche et maintenez le bouton enfoncé. Avec la souris, modifier l'orientation spatiale de ce graphique afin de "voir" sous différents angles les tracés de ces ellipses. Saisissant !

Obtenons maintenant des traces sur cette surface avec quelques plans parallèles au plan de coordonnées yOz . Dans de tels plans, l'abscisse est constant. Posons $x = k$.

```
> Éq_particulière:=subs(x=k,Éq_générale);
```

$$\text{Éq_particulière} := \frac{k^2}{16} + \frac{y^2}{4} + \frac{z^2}{4} = 1$$

(2.8)

```
> Éq_transition:=Éq_particulière+(-k^2/16=-k^2/16);
```

(2.9)

$$\acute{E}q_transition := \frac{y^2}{4} + \frac{z^2}{4} = 1 - \frac{k^2}{16} \quad (2.9)$$

Nous constatons que pour chaque valeur k telle que $-4 \leq k \leq 4$, l'équation précédente représente une équation d'une ellipse (cercle) ou d'un point. Exprimons l'équation $\acute{E}q_transition$ sous la forme réduite.

```
> Équation_ellipse:=(1/(1-k^2/16))*Éq_transition;
expand(Équation_ellipse);
```

$$\acute{E}quation_ellipse := \frac{\frac{y^2}{4} + \frac{z^2}{4}}{1 - \frac{k^2}{16}} = 1$$

$$\frac{y^2}{4 \left(1 - \frac{k^2}{16}\right)} + \frac{z^2}{4 \left(1 - \frac{k^2}{16}\right)} = 1 \quad (2.10)$$

Afin de tracer, à partir des équations paramétriques, chaque ellipse d'équation

$$\frac{y^2}{4 \left(1 - \frac{k^2}{16}\right)} + \frac{z^2}{4 \left(1 - \frac{k^2}{16}\right)} = 1, \text{ posons } a1 = 2 \sqrt{1 - \frac{k^2}{16}} \text{ et } b1 = 2 \sqrt{1 - \frac{k^2}{16}}$$

```
> a1:=2*sqrt(1-k^2/16);
b1:=2*sqrt(1-k^2/16);
```

$$a1 := \frac{\sqrt{-k^2 + 16}}{2}$$


$$b1 := \frac{\sqrt{-k^2 + 16}}{2} \quad (2.11)$$

Construisons quelques traces avec des valeurs $-4 \leq k \leq 4$.

```
> Pas:=[seq(-4+k*0.1,k=0..5),seq(-3+k*0.5,k=0..5),seq(1+k*0.5,k=-2..4),
seq(3.5+k*0.1,k=0..5)];
```


```
Pas := [-4., -3.9, -3.8, -3.7, -3.6, -3.5, -3., -2.5, -2., -1.5, -1., -0.5, 0., 0.5, 1., 1.5, 2.,
2.5, 3., 3.5, 3.6, 3.7, 3.8, 3.9, 4.] \quad (2.12)
```

```
> Ellipse_yOz:=seq(spacecurve([k,a1*cos(t),b1*sin(t)],t=0..2*Pi,
color="Bright 5"),k=Pas):
display(Ellipse_yOz,view=[-5..5,-3..3,-3..3]);
```



Superposons ces deux familles d'ellipses dans un même graphique pour se faire une première idée de la surface.

```
> display(Ellipse_xOz,Ellipse_yOz,view=[-5..5,-3..3,-3..3]);
```



Finalement, obtenons quelques traces avec des plans parallèles au plan de coordonnées xOy . Posons $z = k$.

```
> Éq_particulière:=subs(z=k,Éq_générale);
```

$$\text{Éq_particulière} := \frac{x^2}{16} + \frac{y^2}{4} + \frac{k^2}{4} = 1 \quad (2.13)$$

```
> Éq_transition:=Éq_particulière+(-k^2/4=-k^2/4);
```

$$\text{Éq_transition} := \frac{x^2}{16} + \frac{y^2}{4} = 1 - \frac{k^2}{4} \quad (2.14)$$

Pour chaque valeur k telle que $-2 \leq k \leq 2$, l'équation précédente représente une équation d'une ellipse. Exprimons Éq_transition sous la forme réduite.

```
> Équation_ellipse:=(1/(1-k^2/4))*Éq_transition;
expand(Équation_ellipse);
```

$$\begin{aligned} \text{Équation_ellipse} &:= \frac{\frac{x^2}{16} + \frac{y^2}{4}}{1 - \frac{k^2}{4}} = 1 \\ \frac{x^2}{16 \left(1 - \frac{k^2}{4}\right)} + \frac{y^2}{4 \left(1 - \frac{k^2}{4}\right)} &= 1 \end{aligned} \quad (2.15)$$

Posons alors pour l'ellipse d'équation $\frac{x^2}{16 \left(1 - \frac{k^2}{4}\right)} + \frac{y^2}{4 \left(1 - \frac{k^2}{4}\right)} = 1$, $a1 = 4 \sqrt{1 - \frac{k^2}{4}}$ et

$$b1 = 2 \sqrt{1 - \frac{k^2}{4}}$$

```
> a1:=4*sqrt(1-k^2/4);
  b1:=2*sqrt(1-k^2/4);
```

$$a1 := 2\sqrt{-k^2 + 4}$$

$$b1 := \sqrt{-k^2 + 4}$$


(2.16)

Construisons quelques traces avec des valeurs $-2 \leq k \leq 2$.

```
> Pas:=[-2,-1.9,-1.8,-1.7,-1.6,-1.5,-1,-0.5,0,0.25,0.5,1,1.5,1.6,1.7,
  1.8,1.9,2];
```


```
  Pas := [-2, -1.9, -1.8, -1.7, -1.6, -1.5, -1, -0.5, 0, 0.25, 0.5, 1, 1.5, 1.6, 1.7, 1.8, 1.9, 2] (2.17)
```

```
> Ellipse_xOy:=seq(spacecurve([a1*cos(t),b1*sin(t),k],t=0..2*Pi,
  color="Bright 7"),k=Pas):
  display(Ellipse_xOy,view=[-5..5,-3..3,-3..3],orientation=[55,65,0]);
```




Superposons finalement les trois familles d'ellipses dans un même graphique. Voici donc la forme de cette surface.

```
> display(Ellipse_xOz,Ellipse_yOz,Ellipse_xOy,view=[-5..5,-3..5,-3..3])
  ;
```




Cette surface étant définie implicitement, la macro-commande `implicitplot3d` de l'extension `plots` peut être utilisée pour obtenir directement une représentation de cette surface.

```
> Ellipsoïde:=implicitplot3d(x^2/a^2+y^2/b^2+z^2/c^2 = 1,x=-4..4,y=-2.
.2,z=-2..2,
    grid=[20,20,30],
    style=surface,
    color="CVD 10",
    lightmodel=light1):
display(Ellipsoïde,view=[-5..5,-3..3,-3..3]);
```




Visualisons les différentes ellipses comme traces laissées sur cette surface par des plans parallèles aux plans de coordonnées. Nous allons illustrer, à chaque fois, qu'un seul des plans parallèles aux plans de coordonnées qui ont laissées une trace sur la surface.


```
> Plan_xOz:=plot3d([x,1,z],x=-4..4,z=-4..4,color="Bright 3"):
display(Ellipsoïde,Plan_xOz,Ellipse_xOz,
    lightmodel=none,style=patchngrid,
    orientation=[30,75],
    view=[-4..4,-5..5,-4..4]);
```



```
> Plan_yOz:=plot3d([2,y,z],y=-5..5,z=-4..4,color="Bright 5"):
display(Ellipsoïde,Plan_yOz,Ellipse_yOz,lightmodel=none,view=[-4..4,
-5..5,-4..4],style=patchnogrid,orientation=[70,75]);
```



```
> Plan_xOy:=plot3d([x,y,0.5],x=-5..5,y=-5..5,color="Bright 7"):
display(Ellipsoïde,Plan_xOy,Ellipse_xOy,lightmodel=none,view=[-4..4,
-5..5,-4..4],style=patchnogrid,orientation=[25,75]);
```



Précisons, en terminant, que nous pouvons également tracer cet ellipsoïde directement à l'aide des équations paramétriques suivantes:

$$x = a \cos(v) \cos(u)$$

$$y = b \cos(v) \sin(u)$$

$$z = c \sin(v)$$

Le rendu de la surface est supérieur à celui qui a été obtenu avec la macro-commande `impliciplot3d`.


```
> Surface:=plot3d([a*cos(v)*cos(u),b*cos(v)*sin(u),c*sin(v)],u=0..2*Pi,
  v=-Pi/2..Pi/2,
  grid=[30,30]):
> display(Surface,
  title=typeset("Ellipsoïde d'équations paramétriques:\n ",x=
  a*cos(v)*cos(u),"\n",
  y=b*cos(v)*sin(u),"\n",
  z=c*sin(v)),
  orientation=[50,75],
  lightmodel=light4,
  view=[-5..5,-5..5,-3..3]);
```

Ellipsoïde d'équations paramétriques:

$$x = 4 \cos(v) \cos(u)$$

$$y = 2 \cos(v) \sin(u)$$

$$z = 3 \sin(v)$$



```
> unassign('a,b,c');
```

Hyperboloïde à une nappe d'équation $\frac{x^2}{a^2} + \frac{y^2}{b^2} - \frac{z^2}{c^2} = 1$

Soit l'hyperboloïde à une nappe d'équation $\frac{x^2}{4} + \frac{y^2}{9} - \frac{z^2}{25} = 1$.

```
> unassign('a,b,c');
```

```
Éq_générale:=x^2/a^2+y^2/b^2-z^2/c^2 = 1;
```

$$\text{Éq_générale} := \frac{x^2}{a^2} + \frac{y^2}{b^2} - \frac{z^2}{c^2} = 1 \quad (3.1)$$

```
> a:=2:
```

```
b:=3:
```

```
c:=5:
```

```
Hyperboloïde:=Éq_générale;
```

$$\text{Hyperboloïde} := \frac{x^2}{4} + \frac{y^2}{9} - \frac{z^2}{25} = 1 \quad (3.2)$$

Obtenons quelques les traces laissées sur cet hyperboloïde à une nappe par des plans parallèles aux plans de coordonnées.

Commençons avec des plans parallèles au plan de coordonnées xOz . Posons $y = k$.

```
> Éq_particulière:=subs(y=k,Éq_générale);
```

$$\text{Éq_particulière} := \frac{x^2}{4} + \frac{k^2}{9} - \frac{z^2}{25} = 1 \quad (3.3)$$

```
> Éq_transition:=Éq_particulière+(-k^2/9=-k^2/9);
```

$$\text{Éq_transition} := \frac{x^2}{4} - \frac{z^2}{25} = 1 - \frac{k^2}{9} \quad (3.4)$$

Pour chaque valeur $k \in \mathbb{R}$, (sauf avec $k = \pm 3$) l'équation précédente représente une équation d'une hyperbole (non dégénérée). Nous déduisons aisément que les deux cas d'hyperboles dégénérées sont les droites d'équations $z = \pm \frac{5}{2}x$.

Exprimons Éq_transition sous la forme réduite.

```
> Équation_hyperbole:=(1/(1-k^2/9))*Éq_transition;
expand(Équation_hyperbole);
```

$$\text{Équation_hyperbole} := \frac{\frac{x^2}{4} - \frac{z^2}{25}}{1 - \frac{k^2}{9}} = 1$$

$$\frac{x^2}{4 \left(1 - \frac{k^2}{9}\right)} - \frac{z^2}{25 \left(1 - \frac{k^2}{9}\right)} = 1 \quad (3.5)$$

Dans ce cas-ci, nous devons déployer un peu plus d'efforts pour pouvoir donner une bonne illustration de ces traces. En effet, nous devons traiter deux cas: $-3 < k < 3$ et $k < -3$ ou $k > 3$ puisque le signe du calcul de $1 - \frac{k^2}{9}$ change sur ces intervalles.

Pour $-3 < k < 3$, nous avons la famille d'hyperboles de la forme

$$\frac{x^2}{4\left(1 - \frac{k^2}{9}\right)} - \frac{z^2}{25\left(1 - \frac{k^2}{9}\right)} = 1.$$

Pour $k < -3$ ou $k > 3$, nous avons plutôt la famille d'hyperboles de la forme

$$\frac{z^2}{25\left|1 - \frac{k^2}{9}\right|} - \frac{x^2}{4\left|1 - \frac{k^2}{9}\right|} = 1.$$

Nous allons utiliser des équations paramétriques d'une hyperbole pour mieux obtenir leur tracé.

Des équations paramétriques d'une hyperbole centrée à l'origine d'équation $\frac{x^2}{a^2} - \frac{y^2}{b^2} = 1$ sont

$$\begin{aligned} x &= \frac{a}{\cos(t)} \\ y &= b \tan(t) \end{aligned}$$

Obtenons, en premier, quelques hyperboles avec $-3 < k < 3$ en posant $a1 = 2\sqrt{1 - \frac{k^2}{9}}$ et

$$b1 = 5\sqrt{1 - \frac{k^2}{9}}$$

```
> a1:=2*sqrt(1-k^2/9);
   b1:=5*sqrt(1-k^2/9);
```

$$a1 := \frac{2\sqrt{-k^2 + 9}}{3}$$

$$b1 := \frac{5\sqrt{-k^2 + 9}}{3}$$

(3.6)

Construisons quelques hyperboles avec des valeurs $-3 < k < 3$.


```
> Pas:=[-2.9,-2.5,-2,-1.5,-1,-.5,0,0.5,1,1.5,2.5,2.9];
   Pas := [-2.9, -2.5, -2, -1.5, -1, -0.5, 0, 0.5, 1, 1.5, 2.5, 2.9]
```

(3.7)

Pour mieux illustrer les hyperboles, nous allons les tracer en deux étapes: avec $t = -\frac{\pi}{3} .. \frac{\pi}{3}$ pour des valeurs

positives d'abscisse et avec $t = -\frac{4\pi}{3} .. -\frac{2\pi}{3}$ pour des valeurs négatives d'abscisses.

```
> Hyperbole_xOz1:=seq(spacecurve([a1/cos(t),k,b1*tan(t)],t=-Pi/3..Pi/3,
   color="Bright 3"),k=Pas):
   Hyperbole_xOz2:=seq(spacecurve([a1/cos(t),k,b1*tan(t)],t=-4*Pi/3..-2*
   Pi/3,color="Bright 3"),k=Pas):
> display(Hyperbole_xOz1,Hyperbole_xOz2,scaling=unconstrained);
```



Pour la construction d'hyperboles correspondant à des valeurs $k < -3$ ou des valeurs $k > 3$, nous devons inverser l'abscisse et l'ordonnée dans spacecurve afin d'invertir le rôle des variables x et z .

```
> a1:=5*sqrt(abs(1-k^2/9));
   b1:=2*sqrt(abs(1-k^2/9));
```

$$a1 := 5 \sqrt{\left| -1 + \frac{k^2}{9} \right|}$$


$$b1 := 2 \sqrt{\left| -1 + \frac{k^2}{9} \right|}$$

(3.8)

```
> Pas:=[-6,-5,-4,-3.5,-3.2,-3.1,-3,3,3.1,3.2,3.5,4,5,6];
   Pas := [-6, -5, -4, -3.5, -3.2, -3.1, -3, 3, 3.1, 3.2, 3.5, 4, 5, 6]
```

(3.9)


```
> Hyperbole_xOz3:=seq(spacecurve([b1*tan(t),k,a1/cos(t)],t=-Pi/3..Pi/3,
   color="Bright 5"),k=Pas):
Hyperbole_xOz4:=seq(spacecurve([b1*tan(t),k,a1/cos(t)],t=-4*Pi/3..-2*
   Pi/3,color="Bright 5"),k=Pas):
> display(Hyperbole_xOz3,Hyperbole_xOz4,scaling=unconstrained);
```



Superposons ces hyperboles dans un même graphique.

```
> display(Hyperbole_xOz1,Hyperbole_xOz2,Hyperbole_xOz3,Hyperbole_xOz4,
```

```
orientation=[55,66],
view=[-5..5,-5..5,-10..10],
scaling=unconstrained);
```



Maintenant, obtenons quelques traces avec des plans parallèles au plan de coordonnées yOz . Posons $x = k$.

L'analyse est immédiate par symétrie. Nous allons donc procéder plus rapidement. Le lecteur est tout de même invité à faire une lecture attentive des développements ci-dessous pour voir comment la symétrie a été gérée.

```
> Éq_particulière:=subs(x=k,Éq_générale);
```

$$\text{Éq_particulière} := \frac{k^2}{4} + \frac{y^2}{9} - \frac{z^2}{25} = 1 \quad (3.10)$$

```
> Éq_transition:=Éq_particulière+(-k^2/4=-k^2/4);
```

$$\text{Éq_transition} := \frac{y^2}{9} - \frac{z^2}{25} = 1 - \frac{k^2}{4} \quad (3.11)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, (sauf avec $k = \pm 2$, c'est un point) l'équation précédente représente une équation d'une hyperbole. Exprimons Éq_transition sous la forme réduite.

```
> Équation_hyperbole:=(1/(1-k^2/4))*Éq_transition;
expand(Équation_hyperbole);
```

$$\text{Équation_hyperbole} := \frac{\frac{y^2}{9} - \frac{z^2}{25}}{1 - \frac{k^2}{4}} = 1$$

$$\frac{y^2}{9 \left(1 - \frac{k^2}{4}\right)} - \frac{z^2}{25 \left(1 - \frac{k^2}{4}\right)} = 1 \quad (3.12)$$

Posons alors pour l'hyperbole d'équation $\frac{y^2}{9 \left(1 - \frac{k^2}{4}\right)} - \frac{z^2}{25 \left(1 - \frac{k^2}{4}\right)} = 1$, $a1 = 3 \sqrt{1 - \frac{k^2}{4}}$ et

$$b1 = 5 \sqrt{1 - \frac{k^2}{4}}$$

```
> a1:=3*sqrt(1-k^2/4);
b1:=5*sqrt(1-k^2/4);
```

$$a1 := \frac{3\sqrt{-k^2 + 4}}{2}$$

$$b1 := \frac{5\sqrt{-k^2 + 4}}{2}$$

(3.13)

Construisons quelques traces avec des valeurs $-2 \leq k \leq 2$.

```
> Pas:=[-2,-1.9,-1.5,-1,-0.5,0,0.5,1,1.5,1.9,2];
```


```
Pas := [-2, -1.9, -1.5, -1, -0.5, 0, 0.5, 1, 1.5, 1.9, 2]
```

(3.14)

```
> Hyperbole_yOz1:=seq(spacecurve([k,a1/cos(t),b1*tan(t)],t=-Pi/3..Pi/3,
color="Bright 7"),k=Pas):
```

```
Hyperbole_yOz2:=seq(spacecurve([k,a1/cos(t),b1*tan(t)],t=-4*Pi/3..-2*
Pi/3,color="Bright 7"),k=Pas):
```

```
> display(Hyperbole_yOz1,Hyperbole_yOz2,scaling=unconstrained);
```



Pour la construction d'hyperboles avec des valeurs $k < -2$ ou des valeurs $k > 2$, il faut inverser l'abscisse et l'ordonnée dans `spacecurve` afin d'invertir le rôle des variables x et z .

```
> a1:=5*sqrt(abs(1-k^2/4));
b1:=3*sqrt(abs(1-k^2/4));
```

$$a1 := 5 \sqrt{\left| -1 + \frac{k^2}{4} \right|}$$

$$b1 := 3 \sqrt{\left| -1 + \frac{k^2}{4} \right|}$$

(3.15)

```
> Pas:=[-5,-4,-3.5,-3,-2.5,-2.3,-2.2,-2.1,-2.2,1.2.2,2.4,2.5,3,3.5,4,5]
;
```


```
Pas := [-5, -4, -3.5, -3, -2.5, -2.3, -2.2, -2.1, -2.2, 2.4, 2.4, 2.5, 3, 3.5, 4, 5]
```

(3.16)

```
> Hyperbole_yOz3:=seq(spacecurve([k,b1*tan(t),a1/cos(t)],t=-Pi/3..Pi/3,
color="Bright 9"),k=Pas):
```


```
Hyperbole_yOz4:=seq(spacecurve([k,b1*tan(t),a1/cos(t)],t=-4*Pi/3..-2*
```

```
Pi/3,color="Bright 9"),k=Pas):
> display(Hyperbole_yOz3,Hyperbole_yOz4,scaling=unconstrained);
```




Superposons ces deux familles d'hyperboles dans un même graphique.

```
> display(Hyperbole_xOz1,Hyperbole_xOz2,Hyperbole_xOz3,Hyperbole_xOz4,
orientation=[55,66],
view=[-5..5,-5..5,-10..10]);
```



Superposons, pour voir, toutes les hyperboles déjà tracées.

```
> display(Hyperbole_xOz1,Hyperbole_xOz2,Hyperbole_xOz3,Hyperbole_xOz4,
Hyperbole_yOz1,Hyperbole_yOz2,Hyperbole_yOz3,Hyperbole_yOz4,
orientation=[55,66],
view=[-5..5,-5..5,-10..10]);
```



Obtenons quelques traces avec des plans parallèles au plan de coordonnées xOy . Posons $z = k$.

> $\text{Éq_particulière} := \text{subs}(z=k, \text{Éq_générale});$

$$\text{Éq_particulière} := \frac{x^2}{4} + \frac{y^2}{9} - \frac{k^2}{25} = 1 \quad (3.17)$$

> $\text{Éq_transition} := \text{Éq_particulière} + (k^2/25 = k^2/25);$

$$\text{Éq_transition} := \frac{x^2}{4} + \frac{y^2}{9} = 1 + \frac{k^2}{25} \quad (3.18)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente une équation d'une ellipse. Exprimons Éq_transition sous la forme réduite.

> $\text{Équation_ellipse} := (1/(1+k^2/25)) * \text{Éq_transition};$
 $\text{expand}(\text{Équation_ellipse});$

$$\begin{aligned} \text{Équation_ellipse} &:= \frac{\frac{x^2}{4} + \frac{y^2}{9}}{1 + \frac{k^2}{25}} = 1 \\ &= \frac{\frac{x^2}{4\left(1 + \frac{k^2}{25}\right)} + \frac{y^2}{9\left(1 + \frac{k^2}{25}\right)}}{1} = 1 \end{aligned} \quad (3.19)$$

Posons alors pour l'ellipse d'équation $\frac{x^2}{4\left(1 + \frac{k^2}{25}\right)} + \frac{y^2}{9\left(1 + \frac{k^2}{25}\right)} = 1$, $a1 = 2\sqrt{1 + \frac{k^2}{25}}$ et


$$b1 = 3\sqrt{1 + \frac{k^2}{25}}$$

> $a1 := 2 * \text{sqrt}(1+k^2/25);$
 $b1 := 3 * \text{sqrt}(1+k^2/25);$

$$\begin{aligned} a1 &:= \frac{2\sqrt{k^2 + 25}}{5} \\ b1 &:= \frac{3\sqrt{k^2 + 25}}{5} \end{aligned} \quad (3.20)$$


Construisons quelques traces avec des valeurs $-10 < k < 10$.

```
> Pas:=[-10,-8,-6,-4,-2,1,0,1,2,4,6,8,10];
      Pas := [-10, -8, -6, -4, -2, 1, 0, 1, 2, 4, 6, 8, 10] (3.21)
> Ellipse_xOy:=seq(spacecurve([a1*cos(t),b1*sin(t),k],t=0..2*Pi,
  color="Bright 11"),k=Pas):
> display(Ellipse_xOy);
```



Superposons toutes les hyperboles et cette famille d'ellipses dans un même graphique. Voici donc la forme de cette surface.

```
> display(Hyperbole_xOz1,Hyperbole_xOz2,Hyperbole_xOz3,Hyperbole_xOz4,
  Hyperbole_yOz1,Hyperbole_yOz2,Hyperbole_yOz3,Hyperbole_yOz4,
  Ellipse_xOy,
  orientation=[55,66],
  view=[-6..6,-8..8,-10..10]);
```




Cette surface étant définie implicitement, la macro-commande `implicitplot3d` de l'extension `plots` peut être utilisée pour obtenir directement une représentation de cette surface.

```
> Hyperboloïde:=implicitplot3d(x^2/a^2+y^2/b^2-z^2/c^2 = 1,x=-6..6,y=
  -8..8,z=-10..10,
```

```

grid=[20,20,10],
style=surface,
color="Niagara 4",
lightmodel=light1):
> display(Hyperboloïde,axes=framed);

```




Visualisons, sur la surface précédente, les différentes traces laissées par des plans parallèles aux plans de coordonnées.

```

> display(Hyperboloïde,Hyperbole_xOz1,Hyperbole_xOz2,Hyperbole_xOz3,
Hyperbole_xOz4,
Hyperbole_yOz1,Hyperbole_yOz2,Hyperbole_yOz3,Hyperbole_yOz4,
Ellipse_xOy,
axes=framed,
style=patchnogrid,
transparency=0.6,
orientation=[55,66],
view=[-6..6,-8..8,-10..10]);

```



Pour terminer, précisons que nous pouvons également tracer cet hyperboloïde à une nappe directement à l'aide des équations paramétriques suivantes:


$$\begin{aligned}x &= a(\cos(u) - v\sin(u)) \\y &= b(\sin(u) + v\cos(u)) \\z &= cv\end{aligned}$$

```
> Surface:=plot3d([a*(cos(u)-v*sin(u)),b*(sin(u)+v*cos(u)),c*v],u=-Pi..
Pi,v=-Pi..Pi,
    grid=[20,15],
    title=typeset("Hyperboloïde à une nappe d'équations
paramétriques \n ",x=a*(cos(u)-v*sin(u)),"\n",
    y=b*(sin(u)+v*cos(u)),"\n"
    ,z=c*v),
    lightmodel=light4,
    orientation=[50,75]):
```

Surface;

Hyperboloïde à une nappe d'équations paramétriques


$$\begin{aligned}x &= 2\cos(u) - 2v\sin(u) \\y &= 3\sin(u) + 3v\cos(u) \\z &= 5v\end{aligned}$$



```
> display(Surface,
    Hyperbole_xOz1,Hyperbole_xOz2,Hyperbole_xOz3,Hyperbole_xOz4,
    Hyperbole_yOz1,Hyperbole_yOz2,Hyperbole_yOz3,Hyperbole_yOz4,
    Ellipse_xOy,
    title=typeset("Hyperboloïde à une nappe d'équation:\n
",Éq_générale),
    style=patchnogrid,
    lightmodel=light4,
    transparency=0.6,
    orientation=[55,66],
    view=[-6..6,-8..8,-10..10]);
```

Hyperboloïde à une nappe d'équation:

$$\frac{1}{4}x^2 + \frac{1}{9}y^2 - \frac{1}{25}z^2 = 1$$



```
> unassign('a,b,c');
```

Hyperboloïde à deux nappes d'équation $\frac{x^2}{a^2} - \frac{y^2}{b^2} - \frac{z^2}{c^2} = 1$

Soit l'hyperboloïde à deux nappes d'équation $\frac{x^2}{9} - \frac{y^2}{16} - \frac{z^2}{9} = 1$.

```
> unassign('a,b,c');
```

```
Éq_générale:=x^2/a^2-y^2/b^2-z^2/c^2 = 1;
```

$$\text{Éq_générale} := \frac{x^2}{a^2} - \frac{y^2}{b^2} - \frac{z^2}{c^2} = 1 \quad (4.1)$$

```
> a:=3:
```

```
b:=4:
```

```
c:=3:
```

```
Hyperboloïde2:=Éq_générale;
```

$$\text{Hyperboloïde2} := \frac{x^2}{9} - \frac{y^2}{16} - \frac{z^2}{9} = 1 \quad (4.2)$$

Obtenons quelques traces laissées sur cet hyperboloïde à deux nappes par des plans parallèles aux plans de coordonnées.

Commençons avec quelques plans parallèles au plan de coordonnées yOz afin de saisir pourquoi on dit deux nappes. Posons $x = k$.

```
> Éq_particulière:=subs(x=k,Éq_générale);
```

$$\text{Éq_particulière} := \frac{k^2}{9} - \frac{y^2}{16} - \frac{z^2}{9} = 1 \quad (4.3)$$

```
> Éq_transition1:=Éq_particulière+(-k^2/9=-k^2/9);
```

$$\text{Éq_transition1} := -\frac{y^2}{16} - \frac{z^2}{9} = 1 - \frac{k^2}{9} \quad (4.4)$$

```
> Éq_transition2:=-1*Éq_transition1;
```

(4.5)

$$\acute{E}q_transition2 := \frac{y^2}{16} + \frac{z^2}{9} = -1 + \frac{k^2}{9} \quad (4.5)$$

Seules les valeurs k rendant le calcul $-1 + \frac{k^2}{9}$ positif ou nul sont à retenir.

```
> solve(-1+k^2/9>=0,k);
```

$$(-\infty, -3], [3, \infty) \quad (4.6)$$

Pour chaque valeur k telle que $-\infty < k \leq 3$ et $3 \leq k < \infty$, l'équation précédente représente une équation d'une ellipse (ou un point). Autrement c'est un ensemble vide. Ce qui explique que cette surface possède deux parties.

Exprimons $\acute{E}q_transition$ sous la forme réduite.

```
> \acute{E}quation_ellipse:=(1/(-1+k^2/9))*\acute{E}q_transition2;
expand(\acute{E}quation_ellipse);
```

$$\acute{E}quation_ellipse := \frac{\frac{y^2}{16} + \frac{z^2}{9}}{-1 + \frac{k^2}{9}} = 1$$

$$\frac{y^2}{16 \left(-1 + \frac{k^2}{9} \right)} + \frac{z^2}{9 \left(-1 + \frac{k^2}{9} \right)} = 1 \quad (4.7)$$

Obtenons quelques ellipses avec $-\infty < k < -3$ et $3 < k < \infty$ en posant $a1 = 4 \sqrt{-1 + \frac{k^2}{9}}$ et


$$b1 = 3 \sqrt{-1 + \frac{k^2}{9}}$$

```
> a1:=4*sqrt(-1+k^2/9);
b1:=3*sqrt(-1+k^2/9);
```

$$a1 := \frac{4\sqrt{k^2-9}}{3}$$

$$b1 := \sqrt{k^2-9} \quad (4.8)$$

```
> Pas:=[-12,-10,-9,-8,-7,-6,-5,-4,-3.1,3.1,4,5,6,7,8,9,10,12]:
Ellipse_yOz:=seq(spacecurve([k,a1*cos(t),b1*sin(t)],t=0..2*Pi,color=
"Bright 3"),k=Pas):
display(Ellipse_yOz);
```



Obtenons maintenant quelques traces laissées par des plans parallèles au plan de coordonnées xOz . Posons $y = k$.

> $\text{\texttt{Éq_particulière:=subs(y=k,Éq_générale);}}$

$$\text{\texttt{Éq_particulière}} := \frac{x^2}{9} - \frac{k^2}{16} - \frac{z^2}{9} = 1 \quad (4.9)$$

> $\text{\texttt{Éq_transition:=Éq_particulière+(k^2/16=k^2/16);}}$

$$\text{\texttt{Éq_transition}} := \frac{x^2}{9} - \frac{z^2}{9} = 1 + \frac{k^2}{16} \quad (4.10)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente une équation d'une hyperbole. Exprimons $\text{\texttt{Éq_transition}}$ sous la forme réduite.

> $\text{\texttt{Équation_hyperbole:=(1/(1+k^2/16))*Éq_transition;}}$
 $\text{\texttt{expand(Équation_hyperbole);}}$

$$\text{\texttt{Équation_hyperbole}} := \frac{\frac{x^2}{9} - \frac{z^2}{9}}{1 + \frac{k^2}{16}} = 1$$

$$\frac{x^2}{9\left(1 + \frac{k^2}{16}\right)} - \frac{z^2}{9\left(1 + \frac{k^2}{16}\right)} = 1 \quad (4.11)$$

Posons alors, pour l'hyperbole d'équation $\frac{x^2}{9\left(1 + \frac{k^2}{16}\right)} - \frac{z^2}{9\left(1 + \frac{k^2}{16}\right)} = 1$, $a1 = 3\sqrt{1 + \frac{k^2}{16}}$ et

$$b1 = 3\sqrt{1 + \frac{k^2}{16}}$$

> $\text{\texttt{a1:=3*sqrt(1+k^2/16);}}$

$\text{\texttt{b1:=3*sqrt(1+k^2/16);}}$


$$a1 := \frac{3\sqrt{k^2 + 16}}{4}$$

$$b1 := \frac{3\sqrt{k^2 + 16}}{4} \quad (4.12)$$

Construisons quelques traces avec des valeurs $-8 < k < 8$.

```
> Pas:=[-8,-7,-6,-5,-4,-3,-2,-1,0,1,2,3,4,5,6,7,8];
    Pas := [-8, -7, -6, -5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5, 6, 7, 8] (4.13)
```

```
> Hyperbole_xOz1:=seq(spacecurve([a1/cos(t),k,b1*tan(t)],t=-Pi/3..Pi/3,
color="Bright 5"),k=Pas):
Hyperbole_xOz2:=seq(spacecurve([a1/cos(t),k,b1*tan(t)],t=-4*Pi/3..-2*
Pi/3,color="Bright 5"),k=Pas):
display(Hyperbole_xOz1,Hyperbole_xOz2);
```



Finalement, obtenons quelques traces avec des plans parallèles au plan de coordonnées xOy . Posons $z = k$.

```
> Éq_particulière:=subs(z=k,Éq_générale);
    Éq_particulière :=  $\frac{x^2}{9} - \frac{y^2}{16} - \frac{k^2}{9} = 1$  (4.14)
```

```
> Éq_transition:=Éq_particulière+(k^2/9=k^2/9);
    Éq_transition :=  $\frac{x^2}{9} - \frac{y^2}{16} = 1 + \frac{k^2}{9}$  (4.15)
```

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente une équation d'une hyperbole. Exprimons Éq_transition sous la forme réduite.

```
> Équation_hyperbole:=(1/(1+k^2/9))*Éq_transition;
    expand(Équation_hyperbole);
```

$$\text{Équation_hyperbole} := \frac{\frac{x^2}{9} - \frac{y^2}{16}}{1 + \frac{k^2}{9}} = 1$$

$$\frac{x^2}{9\left(1 + \frac{k^2}{9}\right)} - \frac{y^2}{16\left(1 + \frac{k^2}{9}\right)} = 1 \quad (4.16)$$

Posons alors pour l'hyperbole d'équation $\frac{x^2}{9\left(1+\frac{k^2}{9}\right)} - \frac{y^2}{16\left(1+\frac{k^2}{9}\right)} = 1$, $a1 = 3\sqrt{1+\frac{k^2}{9}}$ et

$$b1 = 4\sqrt{1+\frac{k^2}{9}}$$

```
> a1:=3*sqrt(1+k^2/9);
   b1:=4*sqrt(1+k^2/9);
```

$$a1 := \sqrt{k^2 + 9}$$

$$b1 := \frac{4\sqrt{k^2 + 9}}{3}$$

(4.17)


Construisons quelques traces avec des valeurs $-5 \leq k \leq 5$.

```
> Pas:=[-5,-4,-3,-2,-1,0,1,2,3,4,5];
```

```
     Pas := [-5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5]
```


(4.18)

```
> Hyperbole_xOy1:=seq(spacecurve([a1/cos(t),b1*tan(t),k],t=-Pi/3..Pi/3,
   color="Bright 7"),k=Pas):
Hyperbole_xOy2:=seq(spacecurve([a1/cos(t),b1*tan(t),k],t=-4*Pi/3..-2*
   Pi/3,color="Bright 7"),k=Pas):
display(Hyperbole_xOy1,Hyperbole_xOy2);
```




Superposons toutes les hyperboles et cette famille d'ellipses dans un même graphique. Voici donc la forme de cette surface.

```
> display(Ellipse_yOz,
   Hyperbole_xOz1,Hyperbole_xOz2,
   Hyperbole_xOy1,Hyperbole_xOy2,
   axes=framed,
   orientation=[55,66]);
```




Cette surface étant définie implicitement, la macro-commande `implicitplot3d` de l'extension `plots` peut être utilisée pour obtenir directement une représentation de cette surface.

```
> Hyperboloïde2:=implicitplot3d(x^2/a^2-y^2/b^2-z^2/c^2 = 1,x=-12..12,
y=-16..16,z=-12..12,
    grid=[20,20,15],
    style=surface,
    color="Niagara 4",
    lightmodel=light1):
display(Hyperboloïde2,axes=framed);
```



```
> display(Hyperboloïde2,
    Ellipse_yOz,
    Hyperbole_xOz1,Hyperbole_xOz2,
    Hyperbole_xOy1,Hyperbole_xOy2,
    axes=framed,
    transparency=0.6,
    orientation=[55,66],
```

```
style=patchnogrid);
```



En terminant, précisons que nous pouvons également tracer cet hyperboloïde à deux nappes directement à l'aide des équations paramétriques suivantes:


$$\begin{aligned}x &= \frac{c}{\cos(u)} \\y &= b \tan(u) \sin(v) \\z &= a \tan(u) \cos(v)\end{aligned}$$

Nous allons tracer cette surface définie paramétriquement en deux parties, l'une avec des valeurs positives d'abscisses et la seconde, avec des valeurs négatives.

```
> Surface1:=plot3d([c/cos(u),b*tan(u)*sin(v),a*tan(u)*cos(v)],u=-5*
  Pi/12..5*Pi/12,v=-Pi/2..Pi/2,
  grid=[40,40]):
  Surface2:=plot3d([c/cos(u),b*tan(u)*sin(v),a*tan(u)*cos(v)],u=-17*
  Pi/12..-7*Pi/12,v=-Pi/2..Pi/2,
  grid=[40,40]):
> display(Surface1,Surface2,
  title=typeset("Hyperboloïde à deux nappes d'équations
  paramétriques \n ",x=c/cos(u),"\n",
  y=b*tan(u)*sin(v),"\n",
  z=a*tan(u)*cos(v)),
  lightmodel=light4,
  scaling=unconstrained,
  orientation=[60,60]);
```

Hyperboloïde à deux nappes d'équations paramétriques


$$\begin{aligned}x &= \frac{3}{\cos(u)} \\y &= 4 \tan(u) \sin(v) \\z &= 3 \tan(u) \cos(v)\end{aligned}$$



```
> display(Surface1, Surface2,
          Ellipse_yOz,
          Hyperbole_xOz1, Hyperbole_xOz2,
          Hyperbole_xOy1, Hyperbole_xOy2,
          axes=framed,
          title=typeset("Hyperboloïde à deux nappes d'équation:\n ",
                        Éq_générale),
          style=patchnogrid,
          lightmodel=light4,
          orientation=[55,66],
          view=[-10..10,-16..16,-12..12]);
```

Hyperboloïde à deux nappes d'équation:

$$\frac{1}{9}x^2 - \frac{1}{16}y^2 - \frac{1}{9}z^2 = 1$$



```
> unassign('a,b,c');
```

▼ **Cône elliptique d'équation** $z^2 - \frac{x^2}{a^2} - \frac{y^2}{b^2} = 0$

Soit le cône d'équation $z^2 - \frac{x^2}{9} - \frac{y^2}{16} = 0$.

```
> unassign('a,b');
> Éq_générale:=z^2-x^2/a^2-y^2/b^2 = 0;
```

$$\text{Éq_générale} := z^2 - \frac{x^2}{a^2} - \frac{y^2}{b^2} = 0 \quad (5.1)$$

```
> a:=3:
b:=4:
Cône:=Éq_générale;
```

$$\text{Cône} := z^2 - \frac{x^2}{9} - \frac{y^2}{16} = 0 \quad (5.2)$$

Obtenons quelques traces laissées sur ce cône par des plans parallèles aux plans de coordonnées.

Commençons l'illustration de ces traces avec des plans parallèles au plan de coordonnées xOy . Posons $z = k$.

```
> Éq_particulière:=subs(z=k,Éq_générale);
```

$$\text{Éq_particulière} := k^2 - \frac{x^2}{9} - \frac{y^2}{16} = 0 \quad (5.3)$$

```
> Éq_transition1:=Éq_particulière+(-k^2=-k^2);
```

$$\text{Éq_transition1} := -\frac{x^2}{9} - \frac{y^2}{16} = -k^2 \quad (5.4)$$

```
> Éq_transition2:=-1*Éq_transition1;
```

$$\text{Éq_transition2} := \frac{x^2}{9} + \frac{y^2}{16} = k^2 \quad (5.5)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente une équation d'une ellipse. Exprimons Éq_transition sous la forme réduite.

```
> Équation_ellipse:=1/k^2*Éq_transition2;
expand(Équation_ellipse);
```

$$\text{Équation_ellipse} := \frac{\frac{x^2}{9} + \frac{y^2}{16}}{k^2} = 1$$

$$\frac{x^2}{9k^2} + \frac{y^2}{16k^2} = 1 \quad (5.6)$$

Obtenons quelques tracé d'ellipses avec $-8 \leq k \leq 8$ en posant $a1 = 3k$ et $b1 = 4k$.

```
> a1:=3*k;
b1:=4*k;
```


$$a1 := 3k$$

$$b1 := 4k \quad (5.7)$$

```
> Pas:=seq(2*k/3,k=-24..24,3);
```

$$\text{Pas} := -16, -14, -12, -10, -8, -6, -4, -2, 0, 2, 4, 6, 8, 10, 12, 14, 16 \quad (5.8)$$

```
> Ellipse_xOy:=seq(spacecurve([a1*cos(t),b1*sin(t),k],t=0..2*Pi,color=
"Bright 3"),k=Pas):
display(Ellipse_xOy,scaling=unconstrained);
```



Obtenons maintenant quelques traces laissées sur la surface par des plans parallèles au plan de coordonnées xOz . Posons $y = k$.

```
> Éq_particulière:=subs(y=k,Éq_générale);
```

$$\text{Éq_particulière} := z^2 - \frac{x^2}{9} - \frac{k^2}{16} = 0 \quad (5.9)$$

```
> Éq_transition:=Éq_particulière+(k^2/16=k^2/16);
```

$$\text{Éq_transition} := z^2 - \frac{x^2}{9} = \frac{k^2}{16} \quad (5.10)$$

Pour chaque valeur k où $-\infty < k < \infty$, l'équation précédente représente une équation d'une hyperbole. Exprimons Éq_transition sous la forme réduite.

```
> Équation_hyperbole:=1/(k^2/16)*Éq_transition;
expand(Équation_hyperbole);
```

$$\text{Équation_hyperbole} := \frac{16 \left(z^2 - \frac{x^2}{9} \right)}{k^2} = 1$$

$$\frac{16z^2}{k^2} - \frac{16x^2}{9k^2} = 1 \quad (5.11)$$

Posons alors pour l'hyperbole d'équation $\frac{z^2}{\frac{k^2}{16}} - \frac{x^2}{\frac{9k^2}{16}} = 1$, $a1 = \frac{k}{4}$ et $b1 = \frac{3k}{4}$

```
> a1:=k/4;
b1:=3*k/4;
```

$$a1 := \frac{k}{4}$$

$$b1 := \frac{3k}{4} \quad (5.12)$$


Construisons quelques traces avec des valeurs $-15 \leq k \leq 15$.

```
> Pas:=seq(2*k/3,k=-60..60,3);
```

$$\text{Pas} := -40, -38, -36, -34, -32, -30, -28, -26, -24, -22, -20, -18, -16, -14, -12, \quad (5.13)$$


-10, -8, -6, -4, -2, 0, 2, 4, 6, 8, 10, 12, 14, 16, 18, 20, 22, 24, 26, 28, 30, 32, 34, 36, 38, 40

```
> Hyperbole_xOz1:=seq(spacecurve([b1*tan(t),k,a1/cos(t)],t=-Pi/3..Pi/3,
color="Bright 5"),k=Pas):
Hyperbole_xOz2:=seq(spacecurve([b1*tan(t),k,a1/cos(t)],t=-4*Pi/3..-2*
Pi/3,color="Bright 5"),k=Pas):
display(Hyperbole_xOz1,Hyperbole_xOz2,scaling=unconstrained);
```



Superposons cette famille d'hyperboles et la famille d'ellipses dans un même graphique.

```
> display(Ellipse_xOy,
Hyperbole_xOz1,Hyperbole_xOz2,
scaling=unconstrained,
orientation=[30,80]);
```



Finalement, obtenons quelques traces avec des plans parallèles au plan de coordonnées yOz . Posons $x = k$.

```
> Éq_particulière:=subs(x=k,Éq_générale);
```

$$\text{Éq_particulière} := z^2 - \frac{k^2}{9} - \frac{y^2}{16} = 0 \quad (5.14)$$

```
> Éq_transition:=Éq_particulière+(k^2/9=k^2/9);
```

$$\acute{E}q_transition := z^2 - \frac{y^2}{16} = \frac{k^2}{9} \quad (5.15)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente une équation d'une hyperbole. Exprimons $\acute{E}q_transition$ sous la forme réduite.

```
> Équation_hyperbole:=(1/(k^2/9))*Éq_transition;
expand(Équation_hyperbole);
```

$$\acute{E}quation_hyperbole := \frac{9 \left(z^2 - \frac{y^2}{16} \right)}{k^2} = 1$$

$$\frac{9z^2}{k^2} - \frac{9y^2}{16k^2} = 1 \quad (5.16)$$

Posons alors pour l'hyperbole d'équation $\frac{z^2}{\frac{k^2}{9}} - \frac{y^2}{\frac{9k^2}{16}} = 1$, $a1 = \frac{k}{3}$ et $b1 = \frac{4k}{3}$

```
> a1:=k/3;
b1:=4*k/3;
```

$$a1 := \frac{k}{3}$$


$$b1 := \frac{4k}{3} \quad (5.17)$$

Construisons quelques traces avec des valeurs $-15 \leq k \leq 15$.


```
> Pas:=[seq(2*k/3,k=-36..36,3)];
```

```
Pas := [-24, -22, -20, -18, -16, -14, -12, -10, -8, -6, -4, -2, 0, 2, 4, 6, 8, 10, 12, 14, 16,
18, 20, 22, 24] (5.18)
```

```
> Hyperbole_yOz1:=seq(spacecurve([k,b1*tan(t),a1/cos(t)],t=-Pi/3..Pi/3,
color="Bright 7"),k=Pas):
Hyperbole_yOz2:=seq(spacecurve([k,b1*tan(t),a1/cos(t)],t=-4*Pi/3..-2*
Pi/3,color="Bright 7"),k=Pas):
display(Hyperbole_yOz1,Hyperbole_yOz2,scaling=unconstrained,
orientation=[55,66]);
```




```
> display(Ellipse_xOy,
          Hyperbole_xOz1,Hyperbole_xOz2,
          Hyperbole_yOz1,Hyperbole_yOz2,
          axes=framed,
          scaling=unconstrained,
          orientation=[55,66]);
```




Cette surface étant définie implicitement, la macro-commande `implicitplot3d` de l'extension `plots` peut être utilisée pour obtenir directement une représentation de cette surface.

```
> a:=3:
   b:=4:
> Cône:=implicitplot3d(z^2-x^2/a^2-y^2/b^2 = 0,x=-65..65,y=-65..65,z=
  -16..16,
                        grid=[30,20,20],
                        style=surface,
                        color="Niagara 4",
                        lightmodel=light1):
```

```
display(Cône,axes=framed,scaling=unconstrained,orientation=[50,65]);
```



```
> display(Cône,
  Ellipse_xOy,
  Hyperbole_xOz1,Hyperbole_xOz2,
  Hyperbole_yOz1,Hyperbole_yOz2,
  axes=framed,
  transparency=0.6,
  scaling=unconstrained,
  orientation=[55,66],
  style=patchnogrid);
```



Pour terminer, précisons que nous pouvons également tracer ce cône elliptique directement à l'aide des équations paramétriques suivantes:

$$\begin{aligned}x &= a u \cos(v) \\ y &= b u \sin(v) \\ z &= u\end{aligned}$$

Le rendu de la surface est, de loin, supérieur à celui qui a été obtenu avec la macro-commande `impliciplot3d`.


```
> Surface:=plot3d([a*u*cos(v),b*u*sin(v),u],u=-6..6,v=-2*Pi..2*Pi,
```

```

    grid=[40,40],
    title=typeset("Cône elliptique d'équations paramétriques \n
",x=a*u*cos(v),"\n",
    y=b*u*sin(v),"\n",
    z=u),
    lightmodel=light4):
display(Surface,scaling=unconstrained,orientation=[50,75]);

```

Cône elliptique d'équations paramétriques
 $x = 3u \cos(v)$
 $y = 4u \sin(v)$
 $z = u$




```

> display(Surface,
    Ellipse_xOy,
    Hyperbole_xOz1,Hyperbole_xOz2,
    Hyperbole_yOz1,Hyperbole_yOz2,
    axes=framed,
    title=typeset("Cône elliptique d'équations:\n ",Éq_générale),
    style=patchnogrid,
    lightmodel=light4,
    orientation=[50,70],
    style=patchnogrid,
    transparency=0.6,
    scaling=unconstrained,
    view=[-20..20,-25..25,-6..6]);

```

Cône elliptique d'équations:

$$z^2 - \frac{1}{9}x^2 - \frac{1}{16}y^2 = 0$$



```
> unassign('a,b,c');
```

Paraboloïde elliptique d'équation $z - \frac{x^2}{a^2} - \frac{y^2}{b^2} = 0$

Soit le paraboloïde elliptique d'équation $z^2 - \frac{x^2}{9} - \frac{y^2}{16} = 0$.

```
> unassign('a,b');
```

```
Éq_générale:=z-x^2/a^2-y^2/b^2 = 0;
```

$$\text{Éq_générale} := z - \frac{x^2}{a^2} - \frac{y^2}{b^2} = 0 \quad (6.1)$$

```
> a:=3:
```

```
b:=4:
```

```
Paraboloïde_elliptique:=Éq_générale;
```

$$\text{Paraboloïde_elliptique} := z - \frac{x^2}{9} - \frac{y^2}{16} = 0 \quad (6.2)$$

Obtenons quelques traces laissées sur ce paraboloïde par des plans parallèles aux plans de coordonnées.

Commençons l'illustration de ces traces avec des plans parallèles au plan de coordonnées xOy . Posons $z = k$.

```
> Éq_particulière:=subs(z=k,Éq_générale);
```

$$\text{Éq_particulière} := k - \frac{x^2}{9} - \frac{y^2}{16} = 0 \quad (6.3)$$

```
> Éq_transition1:=Éq_particulière+(-k=-k);
```

$$\text{Éq_transition1} := -\frac{x^2}{9} - \frac{y^2}{16} = -k \quad (6.4)$$

```
> Éq_transition2:=-1*Éq_transition1;
```

$$\text{Éq_transition2} := \frac{x^2}{9} + \frac{y^2}{16} = k \quad (6.5)$$

Pour chaque valeur k telle que $0 \leq k < \text{infinity}$, l'équation précédente représente une équation d'une ellipse (ou d'un seul point). Exprimons Éq transition sous la forme réduite.

```
> Équation_ellipse:=1/k*Éq_transition2;
expand(Équation_ellipse);
```

$$\text{Équation_ellipse} := \frac{\frac{x^2}{9} + \frac{y^2}{16}}{k} = 1$$

$$\frac{x^2}{9k} + \frac{y^2}{16k} = 1 \quad (6.6)$$


Obtenons quelques ellipses avec $0 < k \leq 12$ en posant $a1 = 3\sqrt{k}$ et $b1 = 4\sqrt{k}$

```
> a1:=3*sqrt(k);
b1:=4*sqrt(k);
```

$$a1 := 3\sqrt{k}$$

$$b1 := 4\sqrt{k} \quad (6.7)$$

```
> Pas:=[0.125,0.25,0.5,1,2,3,4,5,6,7,8,9,10,11,12]:
Ellipse_xOy:=seq(spacecurve([a1*cos(t),b1*sin(t),k],t=0..2*Pi,color=
"Bright 3"),k=Pas):
display(Ellipse_xOy,orientation=[35,70],scaling=unconstrained);
```



Obtenons maintenant quelques traces laissées par des plans parallèles au plan de coordonnées xOz . Posons $y = k$.

```
> Éq_particulière:=subs(y=k,Éq_générale);
```

$$\text{Éq_particulière} := z - \frac{x^2}{9} - \frac{k^2}{16} = 0 \quad (6.8)$$

```
> Éq_transition:=Éq_particulière+(x^2/9+k^2/16=x^2/9+k^2/16);
```


$$\text{Éq_transition} := z = \frac{x^2}{9} + \frac{k^2}{16} \quad (6.9)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente une équation d'une parabole.

Construisons quelques paraboles avec des valeurs $-12 < k < 12$.


```
> Pas:= [seq(2*k,k=-5..5)];
      Pas := [-10, -8, -6, -4, -2, 0, 2, 4, 6, 8, 10] (6.10)
```

```
> Parabole_xOz:=seq(spacecurve([x,k,1/9*x^2+1/16*k^2],x=-6..6,color=
"Bright 5"),k=Pas):
display(Parabole_xOz,orientation=[35,70],scaling=unconstrained);
```



Superposons cette famille de paraboles et la famille d'ellipses dans un même graphique.

```
> display(Ellipse_xOy,
Parabole_xOz,
orientation=[35,70],
scaling=unconstrained);
```



Finalement, obtenons quelques traces avec des plans parallèles au plan de coordonnées yOz . Posons $x = k$.

```
> Éq_particulière:=subs(x=k,Éq_générale);
      Éq_particulière :=  $z - \frac{k^2}{9} - \frac{y^2}{16} = 0$  (6.11)
```


```
> Éq_transition:=Éq_particulière+(k^2/9+y^2/16=k^2/9+y^2/16);
```

$$\acute{E}q_transition := z = \frac{k^2}{9} + \frac{y^2}{16} \quad (6.12)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente encore une équation d'une parabole.


Construisons quelques paraboles avec des valeurs $-8 < k < 8$.

```
> Pas:= [seq(k,k=-8..8,2)];
      Pas := [-8, -6, -4, -2, 0, 2, 4, 6, 8] (6.13)
> Parabole_yOz:=seq(spacecurve([k,y,1/9*k^2+1/16*y^2],y=-9..9,color=
"Bright 7"),k=Pas):
display(Parabole_yOz,scaling=unconstrained);
```




Superposons les deux familles de paraboles et la famille d'ellipses dans un même graphique. Voici donc la forme de cette surface.

```
> display(Ellipse_xOy,
      Parabole_xOz,
      Parabole_yOz,
      axes=framed,
      orientation=[55,66],
      scaling=unconstrained);
```




Nous pouvons tracer directement ce paraboloid elliptique avec la macro-commande `plot3d` au lieu d'utiliser la macro-commande `implicitplot3d`.

```
> Paraboloid:=plot3d([x,y,1/9*x^2+1/16*y^2],x=-8..8,y=-10..10,
    style=surface,
    color="Niagara 4",
    lightmodel=light1):
display(Paraboloid,scaling=unconstrained);
```




```
> display(Paraboloid,Ellipse_xOy,Parabole_xOz,Parabole_yOz,
    axes=framed,
    style=patchnogrid,
    transparency=0.6,
    scaling=unconstrained,
    orientation=[55,66]);
```




Nous pouvons tracer ce parabolöide elliptique sur un domaine elliptique. Cela donne une allure plus attendue pour cette surface.

```
> Parabolöide_bis:=plot3d([x,y,1/9*x^2+1/16*y^2],x=-11..11,y=-4/3*sqrt
(121-x^2)..4/3*sqrt(121-x^2),
    grid=[45,45],
    style=surface,
    color="Niagara 4",
    lightmodel=light1):
display(Parabolöide_bis,
    scaling=unconstrained,
    orientation=[30,70]);
```



```
> display(Parabolöide_bis,Ellipse_xOy,Parabole_xOz,Parabole_yOz,
    axes=framed,
    style=patchnogrid,
    lightmodel=light4,
    transparency=0.6,
    scaling=unconstrained,
```

```
orientation=[30,70]);
```



Pour terminer, précisons que nous pouvons tracer ce paraboloid elliptique à l'aide des équations paramétriques suivantes:

$$\begin{aligned}x &= \sqrt{a^2 u} \cos(v) \\y &= \sqrt{b^2 u} \sin(v) \\z &= 2u\end{aligned}$$

Le rendu de la surface est supérieur à celui qui est obtenu avec la macro-commande `impliciplot3d`


```
> Surface:=plot3d([sqrt(a^2*u)*cos(v),sqrt(b^2*u)*sin(v),u],u=0..14,v=
0..2*Pi,
    grid=[40,45]):
display(Surface,
    orientation=[30,70],
    title=typeset("Paraboloïde elliptique d'équations
paramétriques \n ",x=sqrt*(a^2*u)*cos(v)," \n",
    y=sqrt(b^2*u)*sin(v)," \n",
    z=2*u),
    lightmodel=light4,
    scaling=unconstrained);
```

Paraboloïde elliptique d'équations paramétriques

$$x = 9 \sqrt{u} \cos(v)$$

$$y = 4 \sqrt{u} \sin(v)$$


$$z = 2u$$



```
> display(Surface, Ellipse_xOy, Parabole_xOz, Parabole_yOz,
          title=typeset("Paraboloïde elliptique d'équation \n
          ", Éq_générale),
          axes=framed,
          lightmodel=light3,
          style=patchnogrid,
          transparency=0.6,
          scaling=unconstrained,
          orientation=[30,70]);
```

Paraboloïde elliptique d'équation

$$z - \frac{1}{9}x^2 - \frac{1}{16}y^2 = 0$$



```
> unassign('a','b');
```

Paraboloïde hyperbolique d'équation $z - \frac{x^2}{a^2} + \frac{y^2}{b^2} = 0$

Soit le paraboloïde hyperbolique d'équation $z - \frac{x^2}{27} + \frac{y^2}{64} = 0$

```
> unassign('a','b');
Éq_générale:=z-x^2/a^2+y^2/b^2 = 0;
```

$$\acute{E}q_g\acute{e}n\acute{e}r\acute{a}l\acute{e} := z - \frac{x^2}{a^2} + \frac{y^2}{b^2} = 0 \quad (7.1)$$

```
> a:=3*sqrt(3):
b:=8:
```

```
> Paraboloid_hyperbolique:=Éq_générale;
```

$$Paraboloid_hyperbolique := z - \frac{x^2}{27} + \frac{y^2}{64} = 0 \quad (7.2)$$

Obtenons quelques traces laissées sur ce paraboloid hyperbolique par des plans parallèles aux plans de coordonnées.

Commençons l'illustration de ces traces avec des plans parallèles au plan de coordonnées yOz . Posons $x = k$.

```
> Éq_particulière:=subs(x=k,Éq_générale);
```

$$\acute{E}q_particulière := z - \frac{k^2}{27} + \frac{y^2}{64} = 0 \quad (7.3)$$

```
> Éq_transition:=Éq_particulière+(-y^2/64+k^2/27=-y^2/64+k^2/27);
```

$$\acute{E}q_transition := z = -\frac{y^2}{64} + \frac{k^2}{27} \quad (7.4)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente une équation d'une parabole.


Obtenons quelques paraboles avec $-10 < k \leq 10$.

```
> Pas:=[seq(k,k=-10..10)];
```

```
Parabole_yOz:=seq(spacecurve([k,y,-1/64*y^2+1/27*k^2],y=-20..20,
color="Bright 2"),k=Pas):
```

```
Pas := [-10, -9, -8, -7, -6, -5, -4, -3, -2, -1, 0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10] \quad (7.5)
```

```
> display(Parabole_yOz,orientation=[35,70],scaling=unconstrained);
```



Obtenons maintenant quelques traces laissées par des plans parallèles au plan de coordonnées xOz . Posons $y = k$.

```
> Éq_particulière:=subs(y=k,Éq_générale);
```

$$\acute{E}q_particul\grave{e}re := z - \frac{x^2}{27} + \frac{k^2}{64} = 0 \quad (7.6)$$

```
> \acute{E}q\_transition:=\acute{E}q\_particul\grave{e}re+(x^2/27-k^2/64=x^2/27-k^2/64);
```


$$\acute{E}q_transition := z = \frac{x^2}{27} - \frac{k^2}{64} \quad (7.7)$$

Pour chaque valeur k telle que $-\infty < k < \infty$, l'équation précédente représente encore une équation d'une parabole.

Construisons quelques paraboles avec des valeurs $-20 < k < 20$.


```
> Pas:=[seq(2*k,k=-10..10)];
Pas := [-20, -18, -16, -14, -12, -10, -8, -6, -4, -2, 0, 2, 4, 6, 8, 10, 12, 14, 16, 18, 20] (7.8)
```

```
> Parabole_xOz:=seq(spacecurve([x,k,1/27*x^2-1/64*k^2],x=-10..10,color=
"Bright 5"),k=Pas):
display(Parabole_xOz,scaling=unconstrained);
```



Superposons ces deux familles de paraboles dans un même graphique.

```
> display(Parabole_yOz,Parabole_xOz,scaling=unconstrained);
```



Finalement, obtenons quelques traces avec des plans parallèles au plan de coordonnées xOy . Posons $z = k$.

> Éq_particulière:=subs(z=k,Éq_générale);

$$\acute{E}q_particulière := k - \frac{x^2}{27} + \frac{y^2}{64} = 0 \quad (7.9)$$

> Éq_transition1:=Éq_particulière+(-k=-k);

$$\acute{E}q_transition1 := -\frac{x^2}{27} + \frac{y^2}{64} = -k \quad (7.10)$$

> Éq_transition2:=Éq_transition1*(-1/k);
expand(Éq_transition2);

$$\acute{E}q_transition2 := -\frac{-\frac{x^2}{27} + \frac{y^2}{64}}{k} = 1$$

$$\frac{x^2}{27k} - \frac{y^2}{64k} = 1 \quad (7.11)$$

Pour chaque valeur k epsilon $\mathbb{R} \setminus \{0\}$, l'équation précédente représente l'équation d'une hyperbole. Nous devons traiter deux cas selon le signe de k .

Pour $k > 0$, nous avons la famille d'hyperboles de la forme $\frac{x^2}{27k} - \frac{y^2}{64k} = 1$.

Pour $k < 0$, nous avons plutôt la famille d'hyperboles de la forme $\frac{y^2}{64|k|} - \frac{x^2}{27|k|} = 1$.

Nous allons utiliser des équations paramétriques d'une hyperbole pour mieux obtenir leur tracé. Rappelons que nous avons déjà utilisées les équations paramétriques d'une hyperbole centrée à l'origine d'équation $\frac{x^2}{a^2} - \frac{y^2}{b^2} = 1$. Ces équations sont

$$x = \frac{a}{\cos(t)}$$

$$y = b \tan(t)$$

Obtenons, en premier, quelques hyperboles avec $k > 0$ en posant $a1 = \sqrt{27k} = 3\sqrt{3k}$ et $b1 = \sqrt{64k} = 8\sqrt{k}$.

> a1:=3*sqrt(3*k);
b1:=8*sqrt(k);

$$a1 := 3\sqrt{3}\sqrt{k}$$

$$b1 := 8\sqrt{k} \quad (7.12)$$

Construisons quelques hyperboles avec des valeurs $0 < k \leq 80$.


> Pas:=[seq(k/3,k=1..9)];

$$Pas := \left[\frac{1}{3}, \frac{2}{3}, 1, \frac{4}{3}, \frac{5}{3}, 2, \frac{7}{3}, \frac{8}{3}, 3 \right] \quad (7.13)$$

Pour mieux illustrer les hyperboles, nous allons les tracer en deux étapes: avec $t = -\frac{\text{Pi}}{3} .. \frac{\text{Pi}}{3}$ pour des valeurs

positives d'abscisse et avec $t = -\frac{4\pi}{3} \dots -\frac{2\pi}{3}$ pour des valeurs négatives d'abscisses.

```
> Hyperbole_xOy1:=seq(spacecurve([a1/cos(t),b1*tan(t),k],t=-Pi/4..Pi/4,
color="Bright 7"),k=Pas):
Hyperbole_xOy2:=seq(spacecurve([a1/cos(t),b1*tan(t),k],t=-5*Pi/4..-3*
Pi/4,color="Bright 7"),k=Pas):
display(Hyperbole_xOy1,Hyperbole_xOy2,scaling=unconstrained);
```



Construisons quelques hyperboles correspondant à des valeurs négatives de k . Nous devons inverser l'abscisse et l'ordonnée dans `spacecurve` afin d'invertir le rôle des variables x et y .

```
> a1:=8*sqrt(abs(k));
b1:=3*sqrt(abs(3*k));
```


$$a1 := 8\sqrt{|k|}$$

$$b1 := 3\sqrt{3}\sqrt{|k|} \quad (7.14)$$

```
> Pas:=[seq(k/3,k=-12..-1)];
```


$$Pas := \left[-4, -\frac{11}{3}, -\frac{10}{3}, -3, -\frac{8}{3}, -\frac{7}{3}, -2, -\frac{5}{3}, -\frac{4}{3}, -1, -\frac{2}{3}, -\frac{1}{3} \right] \quad (7.15)$$

```
> Hyperbole_xOy3:=seq(spacecurve([b1*tan(t),a1/cos(t),k],t=-Pi/4..Pi/4,
color="Bright 7"),k=Pas):
Hyperbole_xOy4:=seq(spacecurve([b1*tan(t),a1/cos(t),k],t=-5*Pi/4..-3*
Pi/4,color="Bright 7"),k=Pas):
display(Hyperbole_xOy3,Hyperbole_xOy4,scaling=unconstrained);
```




Superposons ces hyperboles dans un même graphique.

```
> display(Hyperbole_xOy1,Hyperbole_xOy2,Hyperbole_xOy3,Hyperbole_xOy4,
          scaling=unconstrained,
          orientation=[55,66]);
```



Superposons les deux familles de paraboles et celle d'hyperboles dans un même graphique. Voici donc la forme de cette surface.

```
> display(Parabole_yOz,Parabole_xOz,Hyperbole_xOy1,Hyperbole_xOy2,
          Hyperbole_xOy3,Hyperbole_xOy4,
          axes=framed,
          orientation=[55,66],
          scaling=unconstrained);
```



Ce n'est pas fini. Nous avons à visualiser la trace que nous obtenons avec le plan $z = 0$. Avec le plan $z = 0$, l'équation de la trace est $-\frac{x^2}{27} + \frac{y^2}{64} = 0$. Dans ce cas, la trace correspond à deux droites concourantes à l'origine.


```
> solve(-x^2/27+y^2/64 = 0, {y});
```

$$\left\{ y = \frac{8\sqrt{3}x}{9} \right\}, \left\{ y = -\frac{8\sqrt{3}x}{9} \right\}$$

(7.16)


Ajoutons au graphique précédent ces deux traces.

```
> Droite1:=spacecurve([x,8*sqrt(3)*x/9,0],x=-10..10,color="Niagara 2");
Droite2:=spacecurve([x,-8*sqrt(3)*x/9,0],x=-10..10,color="Niagara 2");
:
> display(Parabole_yOz,
          Parabole_xOz,
          Hyperbole_xOy1,Hyperbole_xOy2,Hyperbole_xOy3,Hyperbole_xOy4,
          Droite1,Droite2,
          axes=framed,
          style=patchnogrid,
          scaling=unconstrained,
          orientation=[45,60]);
```




Nous pouvons tracer directement ce paraboloid hyperbolique avec la macro-commande `plot3d` (au lieu d'utiliser la macro-commande `implicitplot3d`).

```
> Paraboloid:=plot3d([x,y,x^2/27-y^2/64],x=-10..10,y=-20..20):
  display(Paraboloid,
           style=surface,
           color="Niagara 4",
           lightmodel=light1,
           scaling=unconstrained);
```



```
> display(Paraboloid,Parabole_yOz,Parabole_xOz,Hyperbole_xOy1,
           Hyperbole_xOy2,Hyperbole_xOy3,Hyperbole_xOy4,
           Droite1,Droite2,
           lightmodel=light4,
           style=patchnogrid,
           scaling=unconstrained,
           orientation=[55,66]);
```




Pour terminer, précisons que nous pouvons tracer ce parabolôide elliptique à l'aide des équations paramétriques suivantes:

$$\begin{aligned}x &= a \cdot u \\y &= b \cdot v \\z &= u^2 - v^2\end{aligned}$$

```
> Surface:=plot3d([a*u,b*v,u^2-v^2],u=-3..3,v=-3..3,
  grid=[20,20]):
  display(Surface,orientation=[60,55],
  style=patch,
  color="Niagara 4",
  lightmodel=light1,
  axes=framed,
  title=typeset("Parabolôide hyperbolique d'équations
paramétriques \n ",x=a*u,"\n",
  y=b*v,"\n",
  z=u^2-v^2),
  scaling=unconstrained);
```

Parabolôide hyperbolique d'équations paramétriques

$$\begin{aligned}x &= 3\sqrt{3}u \\y &= 8v \\z &= u^2 - v^2\end{aligned}$$




```

> display(Surface,
    Parabole_yOz,Parabole_xOz,Hyperbole_xOy1,
    Hyperbole_xOy2,Hyperbole_xOy3,Hyperbole_xOy4,
    Droite1,Droite2,
    title=typeset("Paraboloïde hyperbolique d'équation \n ",
    Éq_générale),
    axes=framed,
    lightmodel=light4,
    style=patchnogrid,
    transparency=0.6,
    scaling=unconstrained,
    view=[-8..8,-20..20,-4..4],
    orientation=[50,60]);

```

Paraboloïde hyperbolique d'équation

$$z - \frac{1}{27}x^2 + \frac{1}{64}y^2 = 0$$



```

> unassign('a,b');

```